Graphs of Functions of Two Variables

```
> restart:with(plots):
> setoptions3d(axes=NORMAL,labels=["x","y","z"],orientation=[20,
  70]);
```

We first look at a graphical illustration of the rectangular coordinate system.

```
> coordplot3d(rectangular);
```



The blue plane is the xy-plane (x=0), the green plane is the xz-plane (y=0), and the red plane is the yz=plane (x=0).

A Surface and Its Cross Sections

We look at the surface defined by the function $z = f(x, y) = 4x^2 + y^2$.

```
> z:=4*x^2+y^2;
```

$$z := 4x^2 + y^2$$

```
> p1:=plot3d(z,x=-5..5,y=-10..10):
> display(p1);
```



We can get a "better" view by setting a **view** or "**z**" window from 0 to 100.

```
> p1:=plot3d(z,x=-5..5,y=-10..10,view=0..100):
> display(p1);
```



At first glance, we see a function with nonnegative range. But there are other ways to get an understanding of this function. First, let's look at a cross section of the graph of f with the plane $x = 1$.

```
> p2:=plot3d([1,y,zz],y=-10..10,zz=0..100,color=green):
> display(p1,p2);
```



We see the intersection is a parabola, in fact the parabola $z = y^2 + 4$. We next take a sequence of 10 cross sections of f with x fixed with values of x from -5 to 5 in steps of size 1.

```
> plot3d({seq([i,y,i^2+y^2],i=-5..5)},x=-5..5,y=-10..10);
```



Compare this graph with the grid markings on the original graph. Now let's look at a cross section of the graph of f with the plane $y = 2$.

```
> p3:=plot3d([x,2,zz],x=-5..5,zz=0..100):
> display(p1,p3);
```



The intersection is again a parabola, in fact the parabola $z = 4\,x^2 + 4$. We again take a sequence of 10 cross sections of f with y fixed with values of y from -10 to 10 in steps of size 2.

```
> plot3d({seq([x,2*i,4*x^2+(2*i)^2],i=-5..5)},x=-5..5,y=-10..10);
```



Compare this graph also with the grid markings on the original graph. What about the cross section of the graph of f with the plane $z = 10$.

```
> p4:=plot3d([x,y,16],x=-5..5,y=-10..10):
> display(p1,p4);
```



From this perspective it is a bit hard to tell visually exactly what a cross section is, but since the equation of the cross section is $4\,x^2 + y^2 = 16$, we know it is an ellipse. We use the contourplot3d command to generate cross sections for f for various fixed values of z.

```
> contourplot3d(z,x=-5..5,y=-10..10);
```



Let's look down from the top of the z-axis.

```
> contourplot3d(z,x=-5..5,y=-10..10,orientation=[-90,0]);
```



It is now visually clear that the cross sections are ellipses. If all the ellipses are all collapsed to the xy-plane, we get what is called a contour diagram.

We now consider the related function z = $f(x, y) = 4\ (x-1)^2 + (y+2)^2 + 6$. Similar to what we encountered with functions of one variable, we have a translation of 1 in the direction of the positive x-axis, a translation of 2 in the direction of the negative y-axis, and a vertical translation of 6 in the direction of the positive z-axis.

```
> z:=4*(x-1)^2+(y+2)^2+6;
```

$$z := 4\ (x-1)^2 + (y+2)^2 + 6$$

```
> plot3d(z,x=-4..11,y=-12..8,view=0..100);
```



>

Linear Functions

Linear functions in three variables are of the form z = $f(x, y) = a\,x + b\,y + c$. Let's consider the graph of the linear function $z = 8\,x + 2\,y + 6$.

```
> z:=-3*x-2*y+6;
```

$$z := -3\,x - 2\,y + 6$$

```
> p5:=plot3d(z,x=-4..4,y=-4..4,style=patchnogrid):
  p6:=pointplot3d({[0,0,6],[2,0,0]},connect=true,color=blue,
  thickness=3):
  p7:=pointplot3d({[0,0,6],[0,3,0]},connect=true,color=blue,
  thickness=3):
  p8:=pointplot3d({[0,3,0],[2,0,0]},connect=true,color=blue,
  thickness=3):
  display(p5,p6,p7,p8);
```



The graph is a plane. The lines joining the three intercepts form a triangle that lies within the plane. Below is a plot of the portion of the plane contained within the triangle.

```
> plot3d(z,x=0..2,y=0..3,view=[0..3,0..4,0..7],style=patchnogrid);
```


7
6
5
x
1.5
2.0
2.5
3.0
y
2
3
4

>